

Asbru protocol for the Management of Diabetes Mellitus Type 2 (using 2 parallel lists for treatment) (version 1)

plan Diabetes-Mellitus-Type-2

intentions

achieve overall-state: (glucose-monitoring $\neq$ bad)

plan-body type = sequentially, wait-for-optional-subplans = yes

wait-for one

Diagnostics
Policy

plan Diagnostics

intentions

achieve overall-state: is-known-parameter(glucose-evaluation) *in* NOW

plan-body type = sequentially, wait-for-optional-subplans = yes

wait-for Anamnesis

Anamnesis
Glucose-determination
Risk-inventory

plan Anamnesis

intentions

achieve overall-state: is-known-variable(glucose-determination-needed)

plan-body type = sequentially

wait-for all

do type = sequentially, wait-for-optional-subplans = yes

wait-for Anamnesis-typical-signs

Anamnesis-typical-signs
Anamnesis-olderthan-45

if (**or** (typical-signs = true) (risk-factors = true))

then

glucose-determination-needed $\leftarrow$ true

else

glucose-determination-needed $\leftarrow$ false

plan Anamnesis-typical-signs

intentions

achieve overall-state: is-known-variable(typical-signs)

plan-body type = sequentially

wait-for all

do type = any-order

wait-for all

ask thirst

ask polyuria

ask weight-loss

if (**and** (age > 60) (sex = female))

then

ask pruritis-vulvae

if (**or** (thirst = true) (polyuria = true) (weight-loss = true) (**and** (age > 60) (sex = female) (pruritis-vulvae = true))
)

then

typical-signs ← true

else

typical-signs ← false

plan Anamnesis-olderthan-45

intentions

achieve overall-state: is-known-variable(risk-factors)

conditions

filter-precondition: (and (age > 45) (triennial-visit = true))

plan-body type = sequentially

wait-for all

do type = any-order

wait-for all

ask DMT2-in-1st-grade-relatives

ask hypertension

ask CV-diseases

ask fat-metabolism-problems

ask quetelet-index

ask ethnic-group

if (sex = female)

then

ask DM-in-past

ask newborns-biggerthan-4kg

if (**or** (DMT2-in-1st-grade-relatives = true) (hypertension = true)
(CV-diseases = true) (fat-metabolism-problems = true) (quetelet-index $\geq$
27) (ethnic-group = hindustanian) (**and** (sex = female) (**or** (DM-in-past
= true) (newborns-biggerthan-4kg = true))
)
)

then

risk-factors $\leftarrow$ true

else

risk-factors $\leftarrow$ false

plan Glucose-determination

intentions

achieve overall-state: (or is-known-parameter(fasting-glucose) *in* NOW
is-known-parameter(postprandial-glucose) *in* NOW
)

conditions

filter-precondition: (glucose-determination-needed = true)

plan-body type = sequentially

wait-for all

do type = any-order

wait-for one

Fasting-glucose-test-manual

Non-fasting-glucose-test-manual

if (glucose-evaluation = DMT2)

then

Fasting-glucose-test *in* ([2d, 7d] [-, -] [-, -]

(or plan-state-transition(leave, first, completed(Fasting-glucose-test-manual)) plan-state-transition(leave, first, completed(Non-fasting-glucose-test-manual)))

)

plan Fasting-glucose-test-manual

Fasting glucose test

conditions

activate-mode: manual

plan-body type = any-order

wait-for all

ask glucose-measurement-type

ask fasting-glucose

plan Non-fasting-glucose-test-manual

Non-fasting glucose test

conditions

activate-mode: manual

plan-body type = any-order

wait-for all

ask glucose-measurement-type

ask non-fasting-glucose

plan Fasting-glucose-test

Fasting glucose test

plan-body type = any-order

wait-for all

ask glucose-measurement-type

ask fasting-glucose

plan Risk-inventory

conditions

filter-precondition: (and is-known-parameter(glucose-evaluation) in
NOW
(glucose-evaluation = DMT2))

plan-body type = any-order

wait-for all

ask CH-diseases
ask CH-disease-in-youngertan-60-1st-grade-relatives
ask feeding-habits
ask smoking
ask alcohol-consuming
ask physical-exercise
ask quetelet-index
ask higher-blood-pressure
ask lower-blood-pressure
if (age < 50)
then
 Albumin-test
ask GHb
ask creatinin
Cholesterol-tests
Feet-examination
Retinopathy-investigation in ([-, -] [-, 6mon] [-, -] SELF)

plan Albumin-test

Albumin test

plan-body type = any-order

wait-for one

Albumin-test-manual
Albumin-creatinin-ratio-test-manual

plan Albumin-test-manual

Albumin test

conditions

activate-mode: manual

plan-body

ask albumin-in-urine

plan Albumin-creatinin-ratio-test-manual

Albumin-creatinin ratio test

conditions

activate-mode: manual

plan-body

ask albumin-creatinin-ratio-in-urine

plan Cholesterol-tests

Cholesterol tests

plan-body type = sequentially

wait-for all

ask total-cholesterol

ask HDL-cholesterol

ask triglycerids

plan Feet-examination

Perform a feet examination

plan-body user-performed

plan Retinopathy-investigation

Perform a retinopathy investigation

plan-body user-performed

plan Policy

intentions

avoid-bad-glucose-values: **avoid** **intermediate-state:**
(glucose-monitoring = bad)

conditions

filter-precondition: (glucose-evaluation = DMT2)

plan-body type = sequentially

wait-for all

Education-DMT2

do type = unordered, retry-aborted-subplans = yes

wait-for Treatments-and-Controls

Treatments-and-Controls

Policy-for-concurrent-diseases

Policy-for-hypoglycemic-coma

Policy-for-consultation

Policy-for-chiropracist-referral

Policy-for-nurse-referral

plan Education-DMT2

Give information about the most important aspects of DM

plan-body user-performed

plan Treatments-and-Controls

intentions

maintain-acceptable-glucose-values: **maintain**
intermediate-state: (**or** (glucose-monitoring = good) (glucose-monitoring
= acceptable))

maintain intermediate-action: activated(Quarterly-control) *in* quarterly

maintain intermediate-action: activated(Annual-control) *in* yearly

avoid-non-desired-situations: **avoid intermediate-state:**
(**or** (hypoglycemic-coma = true) (complications = true))

conditions

complete-condition: (alive = false)

plan-body type = parallel

wait-for none

Non-insulin-DMT2-treatments **on-abort** Insulin-DMT2-treatments
Treatment-of-CV-disease-risk-factors

cyclical-plan

Quarterly-control

retry-delay: min = 10w, max = 14w

cyclical-plan

Annual-control

retry-delay: min = 46w, max = 50w

plan Non-insulin-DMT2-treatments

intentions

achieve-acceptable-glucose-values: **achieve overall-state:**
(**or** (glucose-monitoring = good) (glucose-monitoring = acceptable))

avoid-insulin-treatment: **avoid intermediate-state:** (**or**
(DMT2-treatment = insulin-plus-antidiabetics) (DMT2-treatment =
insulin))

conditions

abort-condition: (**and** (glucose-monitoring = bad) completed(SU-
derivative-plus-metformin-treatment) *in* NOW
)

plan-body type = sequentially, wait-for-optional-subplans = yes

wait-for Diet-specialist-referral

Diet-specialist-referral
Fasting-glucose-test *in* ([11w, 13w] [-, -] [-, -] plan-state-transition(leave,
first, activated(Diet-specialist-referral)))

SU-derivative-or-metformin-treatment
SU-derivative-plus-metformin-treatment

plan Diet-specialist-referral

Diet specialist referral

intentions

avoid intermediate-state: (DMT2-treatment = antidiabetics)
achieve-acceptable-glucose-values^a

plan-body user-performed

^aSee plan Non-insulin-DMT2-treatments.

plan SU-derivative-or-metformin-treatment

variables

antidiabetics: list of String
antidiabetic-doses: list of amount
iterator-antidiabetics: iterator(antidiabetics, first-element)
iterator-antidiabetic-doses: iterator(antidiabetic-doses, first-element)

intentions

avoid intermediate-state: (**or** (gastrointestinal-side-effects = true) (**and**
(hypoglycemia = true) (weight-gain = true))
)

achieve-acceptable-glucose-values^a

conditions

filter-precondition: (glucose-monitoring = bad)

plan-body type = sequentially

wait-for all

DMT2-treatment $\leftarrow$ antidiabetics
if (quetelet-index $\leq$ 27)
then
 ask drug-name
 put-last(drug-name, antidiabetics)
else
 put-last(metformin, antidiabetics)
antidiabetic-doses $\leftarrow$ Initialise-drug-doses(antidiabetics, iterator-
antidiabetics, antidiabetic-doses, iterator-antidiabetic-doses)
antidiabetic-doses $\leftarrow$ Find-antidiabetic-doses(antidiabetics, iterator-
antidiabetics, antidiabetic-doses, iterator-antidiabetic-doses)

^aSee plan Non-insulin-DMT2-treatments.

plan Initialise-drug-doses

arguments

drugs: list of String
iterator-drugs: iterator
drug-doses: list of amount
iterator-drug-doses: iterator

plan-body type = sequentially

wait-for all

reset-iterator(iterator-drugs)
reset-iterator(iterator-drug-doses)
go-to-next(iterator-drug-doses)
do-repeatedly
 ask drug-dose
 insert-before-iterator(drug-dose, iterator-drug-doses)
 go-to-next(iterator-drugs)
 go-to-next(iterator-drug-doses)
termination-condition: is-at-end(iterator-drugs)

returns drug-doses

plan Find-antidiabetic-doses

arguments

drugs: list of String
iterator-drugs: iterator
drug-doses: list of amount
iterator-drug-doses: iterator
antidiabetic-maximal-doses: Boolean

conditions

suspend-condition: (and antidiabetic-problems^a
activated(Check-for-antidiabetic-problems) *in* NOW
)

reactivate-condition: completed(Check-for-antidiabetic-problems) *in*
NOW

complete-condition: (glucose-monitoring $\neq$ bad)

abort-condition: (and (antidiabetic-maximal-doses = true)
(glucose-monitoring = bad))

plan-body

cyclical-plan

drug-doses, antidiabetic-maximal-doses $\leftarrow$ Increase-drug-doses(drugs,
iterator-drugs, drug-doses, iterator-drug-doses)

retry-delay: min = 2w, max = 4w

returns drug-doses

^aSee plan Check-for-antidiabetic-problems.

plan Increase-drug-doses

arguments

drugs: list of String
iterator-drugs: iterator
drug-doses: list of amount
iterator-drug-doses: iterator

variables

aux-doses: list of amount

plan-body type = sequentially

wait-for all

ask maximal-drug-doses

reset-iterator(iterator-drugs)

reset-iterator(iterator-drug-doses)

do-repeatedly

ask drug-dose-increase

put-last((+ drug-dose-increase **get**(iterator-drug-doses)), aux-doses)

go-to-next(iterator-drugs)

go-to-next(iterator-drug-doses)

termination-condition: is-at-end(iterator-drugs)

returns aux-doses, maximal-drug-doses

plan SU-derivative-plus-metformin-treatment

variables

antidiabetics: list of String
antidiabetic-doses: list of amount
iterator-antidiabetics: iterator(antidiabetics, first-element)
iterator-antidiabetic-doses: iterator(antidiabetic-doses, first-element)

intentions

achieve-acceptable-glucose-values^a
avoid-insulin-treatment^b

conditions

filter-precondition: (glucose-monitoring = bad)

plan-body type = sequentially

wait-for all

ask drug-name
put-last(drug-name, antidiabetics)
put-last(metformin, antidiabetics)
antidiabetic-doses ← Initialise-drug-doses(antidiabetics, iterator-
antidiabetics, antidiabetic-doses, iterator-antidiabetic-doses)
do type = unordered
wait-for Find-antidiabetic-doses
antidiabetic-doses ← Find-antidiabetic-doses(antidiabetics, iterator-
antidiabetics, antidiabetic-doses, iterator-antidiabetic-doses)
antidiabetics, antidiabetic-doses ← Check-for-antidiabetic-
problems(antidiabetics, iterator-antidiabetics, antidiabetic-doses,
iterator-antidiabetic-doses)

^aSee plan Non-insulin-DMT2-treatments.

^bSee plan Non-insulin-DMT2-treatments.

plan Check-for-antidiabetic-problems

arguments

drugs: list of String
iterator-drugs: iterator
drug-doses: list of amount
iterator-drug-doses: iterator

conditions

filter-precondition: antidiabetic-problems: **(or**
(contraindications = true) (side-effects = true))

plan-body type = sequentially

wait-for all

ask problematic-antidiabetic
reset-iterator(iterator-drugs)
reset-iterator(iterator-drug-doses)
do-repeatedly
 go-to-next(iterator-drugs)
 go-to-next(iterator-drug-doses)
termination-condition: **(or** **(get**(iterator-drugs) **=**
problematic-antidiabetic) **is-at-end**(iterator-drugs))

remove-at-iterator(iterator-drugs)
remove-at-iterator(iterator-drug-doses)
put-last(acarbose, drugs)
ask drug-dose
put-last(drug-dose, drug-doses)

returns drugs, drug-doses

plan Insulin-DMT2-treatments

intentions

achieve-acceptable-glucose-values^a
avoid-non-desired-situations^b

conditions

filter-precondition: (glucose-monitoring = bad)

plan-body type = sequentially

wait-for all

Education-insulin

do type = any-order

wait-for one

Insulin-with-or-without-antidiabetics-treatment

Only-insulin-treatment

^aSee plan Non-insulin-DMT2-treatments.

^bSee plan Treatments-and-Controls.

plan Education-insulin

Give information about insulin self-control

plan-body user-performed

plan Insulin-with-or-without-antidiabetics-treatment

variables

insulin-dd: amount
insulin-type: insulin-type-scale
evening-insulin-dose: amount
evening-insulin-intake: insulin-intake-scale
morning-insulin-dose: amount
morning-insulin-intake: insulin-intake-scale

intentions

avoid intermediate-state: (insulin-dd > 1)
avoid-hypoglycemic-coma: **avoid** **intermediate-state:**
(hypoglycemic-coma = true)
achieve-acceptable-glucose-values^a

conditions

activate-mode: manual

plan-body type = sequentially, wait-for-optional-subplans = yes

wait-for Four-points-day-curve

DMT2-treatment $\leftarrow$ insulin-plus-antidiabetics
Four-points-day-curve
insulin-type $\leftarrow$ mid-term
insulin-dd $\leftarrow$ 1
ask par-evening-insulin-dose
evening-insulin-dose $\leftarrow$ par-evening-insulin-dose
evening-insulin-intake $\leftarrow$ after-meal
evening-dose $\leftarrow$ Find-evening-insulin-dose(evening-dose) *in* ([2d, 3d] [-,
-] [-, -] SELF)

insulin-dd, evening-insulin-intake, morning-insulin-dose, morning-insulin-
intake $\leftarrow$ Change-to-only-insulin-treatment(evening-insulin-dose)

^aSee plan Non-insulin-DMT2-treatments.

plan Four-points-day-curve

plan-body type = sequentially

wait-for all

ask fasting-glucose
ask after-breakfast-glucose
ask after-lunch-glucose
ask after-dinner-glucose

plan Find-evening-insulin-dose

arguments

evening-dose: amount

conditions

complete-condition: (fasting-glucose-monitoring = good)

abort-condition: (evening-dose $\geq$ 40)

plan-body

cyclical-plan

evening-dose $\leftarrow$ Test-glucose-and-adapt-evening-insulin(evening-dose)

retry-delay: min = 7d, max = 10d

returns evening-dose

plan Test-glucose-and-adapt-evening-insulin

arguments

evening-dose: amount

plan-body type = sequentially

wait-for all

Four-points-day-curve

ask evening-adaptation

returns (+ evening-dose evening-adaptation)

plan Change-to-only-insulin-treatment

arguments

evening-dose: amount

variables

dd: amount
evening-intake: insulin-intake-scale
morning-dose: amount
morning-intake: insulin-intake-scale

intentions

avoid intermediate-state: antidiabetic-problems^a

avoid intermediate-action: activated(Diabetes-specialist-referral) *in*
NOW

achieve-acceptable-glucose-values^b

avoid-hypoglycemic-coma^c

conditions

filter-precondition: (or (and (evening-dose > 40)
(fasting-glucose-monitoring ≠ bad))
(postprandial-glucose-monitoring = bad))

plan-body type = sequentially, wait-for-optional-subplans = yes

wait-for Find-morning-insulin-dose

DMT2-treatment ← insulin
dd ← 2
evening-intake ← before-meal
ask par-morning-insulin-dose
morning-dose ← par-morning-insulin-dose
morning-intake ← before-meal
morning-dose ← Find-morning-insulin-dose(morning-dose)
Diabetes-specialist-referral

returns dd, evening-intake, morning-dose, morning-intake

^aSee plan Check-for-antidiabetic-problems.

^bSee plan Non-insulin-DMT2-treatments.

^cSee plan Insulin-with-or-without-antidiabetics-treatment.

plan Find-morning-insulin-dose

arguments

morning-dose: amount

conditions

complete-condition: (postprandial-glucose-monitoring = good)

plan-body

cyclical-plan

morning-dose $\leftarrow$ Test-glucose-and-adapt-morning-insulin(morning-dose)

retry-delay: min = 7d, max = 10d

returns morning-dose

plan Test-glucose-and-adapt-morning-insulin

arguments

morning-dose: amount

plan-body type = sequentially

wait-for all

Four-points-day-curve

ask morning-adaptation

returns (+ morning-dose morning-adaptation)

plan Diabetes-specialist-referral

Diabetes specialist referral for change of insulin regime

intentions

achieve-acceptable-glucose-values^a

avoid-hypoglycemic-coma^b

conditions

filter-precondition: (postprandial-glucose-monitoring = bad)

plan-body user-performed

^aSee plan Non-insulin-DMT2-treatments.

^bSee plan Insulin-with-or-without-antidiabetics-treatment.

plan Only-insulin-treatment

variables

insulin-dd: amount
insulin-type: insulin-type-scale
evening-insulin-dose: amount
evening-insulin-intake: insulin-intake-scale
morning-insulin-dose: amount
morning-insulin-intake: insulin-intake-scale

intentions

avoid intermediate-state: antidiabetic-problems^a
achieve-acceptable-glucose-values^b
avoid-hypoglycemic-coma^c

conditions

activate-mode: manual

plan-body type = sequentially

wait-for all

DMT2-treatment ← insulin
ask par-insulin-type
insulin-type ← par-insulin-type
insulin-dd ← 2
morning-insulin-dose ← 12
morning-insulin-intake ← before-meal
evening-insulin-dose ← 6
evening-insulin-intake ← before-meal
evening-insulin-dose, morning-insulin-dose ← Find-insulin-
doses(evening-insulin-dose, morning-insulin-dose)

^aSee plan Check-for-antidiabetic-problems.

^bSee plan Non-insulin-DMT2-treatments.

^cSee plan Insulin-with-or-without-antidiabetics-treatment.

plan Find-insulin-doses

arguments

morning-dose: amount
evening-dose: amount

conditions

complete-condition: (glucose-monitoring $\neq$ bad)

plan-body

cyclical-plan

evening-dose, morning-dose $\leftarrow$ Test-glucose-and-adapt-insulin(evening-dose, morning-dose)

retry-delay: min = 2d, max = 3d

returns evening-dose, morning-dose

plan Test-glucose-and-adapt-insulin

arguments

evening-dose: amount
morning-dose: amount

plan-body type = sequentially

wait-for all

Four-points-day-curve
ask evening-adaptation
ask morning-adaptation

returns (+ evening-dose evening-adaptation), (+ morning-dose morning-adaptation)

plan Treatment-of-CV-disease-risk-factors

plan-body type = unordered, wait-for-optional-subplans = yes

wait-for none

Smoking-advice
Hypertension-treatment
Cholesterol-treatment
Microalbuminuria-treatment

plan Smoking-advice

Recommend strongly that the patient stops smoking

conditions

filter-precondition: (smoking = true)

plan-body user-performed

plan Hypertension-treatment

conditions

filter-precondition: (or (higher-blood-pressure > 150)
(lower-blood-pressure > 85))

plan-body

NHG-hypertension-protocol-referral

plan NHG-hypertension-protocol-referral

Refer to NHG Hypertension protocol

plan-body user-performed

plan Cholesterol-treatment

conditions

filter-precondition: (and (fat-metabolism-problems = true)
(life-expectancy > 5) (CH-disease-risk = significant))

plan-body type = sequentially

wait-for all

cholesterol-treatment ← true
NHG-cholesterol-protocol-referral

plan NHG-cholesterol-protocol-referral

Refer to NHG Cholesterol protocol

plan-body user-performed

plan Microalbuminuria-treatment

variables

ACE-inhibitors: list of String
ACE-inhibitors-doses: list of amount
iterator-ACE-inhibitors: iterator(antidiabetics, first-element)
iterator-ACE-inhibitor-doses: iterator(antidiabetic-doses, first-element)

conditions

filter-precondition: (and (age < 50) (microalbuminuria = true))

plan-body type = sequentially

wait-for all

ask drug-name
put-last(drug-name, ACE-inhibitors)
ACE-inhibitor-doses $\leftarrow$ Initialise-drug-doses(ACE-inhibitors, iterator-
ACE-inhibitors, ACE-inhibitor-doses, iterator-ACE-inhibitor-doses)
ACE-inhibitor-doses $\leftarrow$ Find-ACE-inhibitor-doses(ACE-inhibitors,
iterator-ACE-inhibitors, ACE-inhibitor-doses, iterator-ACE-inhibitor-
doses)

plan Find-ACE-inhibitor-doses

arguments

drugs: list of String
iterator-drugs: iterator
drug-doses: list of amount
iterator-drug-doses: iterator
ACE-inhibitor-maximal-doses: Boolean

conditions

complete-condition: (or (and (higher-blood-pressure $\leq$ 140)
(lower-blood-pressure $\leq$ 80))
(ACE-inhibitor-maximal-doses = true))

plan-body

cyclical-plan

drug-doses, ACE-inhibitor-maximal-doses $\leftarrow$ Increase-drug-doses(drugs,
iterator-drugs, drug-doses, iterator-drug-doses)

retry-delay: min = 1mon, max = 3mon

returns drug-doses

plan Quarterly-control

plan-body type = any-order

wait-for all

ask wellbeing

ask hypoglycemia

ask feeding-problems

ask treatment-problems

ask weight

if (**or** (DMT2-treatment = insulin) (DMT2-treatment =
insulin-plus-antidiabetics))

then

Four-points-day-curve

else

Fasting-glucose-test

do type = sequentially

wait-for all

ask feet-problems

if (feet-problems = true)

then

Feet-examination

plan Annual-control

plan-body type = any-order

wait-for all

Quarterly-control

ask visus-problems

ask CV-problems

ask neuropathy

ask sexual-problems

ask higher-blood-pressure

ask lower-blood-pressure

Feet-examination

if (**or** (DMT2-treatment = insulin) (DMT2-treatment =
insulin-plus-antidiabetics))

then

Injection-points-investigation

ask GHb

ask creatinin

if (cholesterol-treatment = false)

then

Cholesterol-tests

if (age < 50)

then

Albumin-test

Funduscopy

plan Injection-points-investigation

Perform an investigation of injection points

plan-body user-performed

plan Funduscopy

Perform a funduscopy

plan-body user-performed

plan Policy-for-concurrent-diseases

conditions

filter-precondition: concurrent-diseases: (**or** (fever = true)
(vomiting = true) (diarrhea = true))

abort-condition: (concurrent-diseases-over = true)

plan-body type = sequentially

wait-for all

Extra-fluid-intake-prescription
Glucose-lowering-therapy-advice
ask concurrent-diseases-over

plan Extra-fluid-intake-prescription

Prescribe extra fluid intake

plan-body user-performed

plan Glucose-lowering-therapy-advice

*Temporary increase based on blood glucose levels. Never discontinue insulin,
even if less food is consumed*

plan-body user-performed

plan Policy-for-hypoglycemic-coma

conditions

filter-precondition: (hypoglycemic-coma = true)

abort-condition: (hypoglycemic-coma-over = true)

plan-body type = sequentially

wait-for all

do type = any-order
wait-for one
Glucose-solution-prescription-manual
Glucagon-prescription-manual
Carbohydrate-rich-feeding-prescription
Hypoglycemia-cause-investigation
ask hypoglycemic-coma-over

plan Glucose-solution-prescription-manual

Prescribe glucose solution

conditions

activate-mode: manual

plan-body user-performed

plan Glucagon-prescription-manual

Prescribe glucagon

conditions

activate-mode: manual

plan-body user-performed

plan Carbohydrate-rich-feeding-prescription

Prescribe carbohydrate rich feeding

conditions

filter-precondition: (conscious = true)

plan-body user-performed

plan Hypoglycemia-cause-investigation

Perform a investigation of hypoglycemia cause

plan-body user-performed

plan Policy-for-consultation

conditions

filter-precondition: (**or** (insulin-adjustment-needed = true) (**and** (postprandial-glucose-monitoring = bad) (insulin-dd = 2)) (diabetes-ulcus-after-2-weeks = true) (creatinin > 200) (creatinin-clearance < 30) (**or** (lethargy = true) (dehydration = true) (vomiting = true)) (hypoglycemic-coma-after-30-min = true))

abort-condition: (consultation-over = true)

plan-body type = sequentially

wait-for all

Consultation

ask consultation-over

plan Consultation

Consultation

plan-body user-performed

plan Policy-for-chiropodist-referral

conditions

filter-precondition: (**or** (pressure-points = true) (bad-standing-habits = true))

abort-condition: (chiropodist-referral-over = true)

plan-body type = sequentially

wait-for all

Chiropodist-referral

ask chiropodist-referral-over

plan Chiropodist-referral

Chiropodist referral

plan-body user-performed

plan Policy-for-nurse-referral

conditions

filter-precondition: (or (self-monitoring-training-needed = true)
(insulin-treatment-start = true))

abort-condition: (nurse-referral-over = true)

plan-body type = sequentially

wait-for all

Nurse-referral

ask nurse-referral-over

plan Nurse-referral

Nurse referral

plan-body user-performed

File created by Mar Marcos on -.

File updates:

- Mar Marcos, 26.2.2002: Changes to solve open issues (after CBO experts)
- Mar Marcos, 23.5.2002: Introduction of treatment intentions (after CBO)
- Mar Marcos, 24.5.2002: Last main update
- Mar Marcos, 28.8.2002: Last small update